Robust Statistics—How Not to Reject Outliers

Part 1. Basic Concepts

Analytical Methods Committee*
Royal Society of Chemistry, Burlington House, Piccadilly, London W1V 0BN, UK

The subject of outliers has been controversial whenever analytical data have been processed. Modern statistical theory provides an alternative to outlier rejection, in which outlying observations are retained but given less weight. This approach is known as robust statistics and is beginning to find favour with analytical chemists. An introduction to robust statistics is given and some examples are described.

Keywords: *Outlier; robust statistics; mean; median; variance*

The Analytical Methods Committee has received and has approved for publication the following report from its Statistical Sub-Committee.

Report

The constitution of the Sub-committee responsible for the preparation of this report was: Dr. M. Thompson (Chairman), Mr. H. M. Bee, Dr. W. H. Evans, Mr. M. J. Gardner, Dr. E. J. Greenhow, Dr. R. Howarth, Dr. E. J. Newman, Professor B. D. Ripley and Dr. R. Wood with Mr. J. J. Wilson as Secretary.

The Analytical Methods Committee gratefully acknowledges the financial support given by the Ministry of Agriculture, Fisheries and Food to the work of this Sub-Committee.

Introduction

Occasionally sets of analytical data occur in which a few observations appear discordant with the remainder. Such observations are known as outliers. For example, considering the following 24 determinations of copper ($\mu g\,g^{-1}$) in wholemeal flour

2.9 3.1 3.4 3.4 3.70 3.7 2.8 2.5 2.4 2.4 2.7 2.2
5.28 3.37 3.03 3.03 28.95 3.77 3.4 2.2 3.5 3.6 3.7 3.7 (1)

one value, 28.95, stands out from the remainder. In this instance we may be particularly suspicious of the value, as a simple explanation suggests itself. Although recording and range errors are almost certainly the major cause of outliers, mistakes can also occur in many other parts of the analytical process and from contamination and transposition of specimens.

The almost universal practice amongst analytical chemists has been to regard outliers as errors, and to delete them from the set of data. In some circumstances this is plainly wrong, and in others there are much safer procedures. Why should we be interested in outliers? One good reason is to catch transcription errors while the original laboratory records are easily accessible. In such an instance we would want to check *all* the extreme results whether or not they are rejected by an outlier test (such as the tests of Dixon[1] or Grubbs[2]). The traditional procedure[3] for dataset (1) would be to compute

$$\max\left[\frac{x(3)-x(1)}{x(22)-x(1)}, \frac{x(24)-x(22)}{x(24)-x(3)}\right] = 0.948$$



Fig. 1. Two views of dataset (1) from the statistical package MINITAB. (*a*) A dot plot; and (*b*) a box plot. * and ○, extreme observations; +, the median; and □, the quantiles

where $x(1)$, ..., $x(24)$ are the observations sorted into increasing order:

2.2 2.2 2.4 2.4 2.5 2.7 2.8 2.9 3.03 3.03 3.1 3.37
3.4 3.4 3.4 3.5 3.6 3.7 3.7 3.7 3.70 3.77 5.28 28.95

and therefore reject $x(24) = 28.95$. For the remaining 23 observations the Dixon[1] test statistic is 0.549, so $x(23) = 5.28$ would also be rejected. Using the test yet again gives 0.133, this being judged not significant. Grubbs' test gives the same results.

The traditional procedure has the merit of pointing out the second outlier, 5.28, but this would have been obvious from any plot of the data (Fig. 1). The second-largest value will be significant (at 5%) only if it exceeds 4.80. However, surely we would want to check a value of 4.77 for a transcription error? If the purpose of detecting outliers is to check the values, it is their extremeness and the plausibility of simple explanations that should weigh in the decision, and *not* the statistical significance.

Outlier rejection is positively wrong when included in a procedure to assess the variability of an analytical method. The outlier rejection procedure used above is that of BS5497[4]; the illustrative dataset (1) is taken from a co-operative trial.[5] The mean and variance of the whole set are 4.28 and 28.1, respectively, whereas after outlier rejection they are 3.11 and 0.281, respectively. If the second-largest observation had been 4.77 we would have obtained 3.19 and 0.387, respectively. From this, two conclusions can be drawn that are true generally. The traditional procedure is (a) sensitive to the actual data values and (b) seriously underestimates the variance that is attainable in practice. The outliers in our dataset are only revealed because we have 24 replicates. Duplication might throw doubt on a value of 28.95, but it is very unlikely to do so on 5.28. On the other hand, estimating the variance by 28.1 is also unfair, as values as large as 28.95 might be spotted and are much rarer than 1 in 24.

* Correspondence should be addressed to the Secretary, Analytical Methods Committee, Analytical Division, Royal Society of Chemistry, Burlington House, Piccadilly, London W1V 0BN, UK.

Barnett and Lewis[6] discussed the outlier problem in considerable detail and described a whole battery of outlier rejection tests. The change of emphasis from their first edition (1978) to the second edition (1984) reflects a change in statistical practice from outlier *rejection* to outlier *accommodation*. The prevailing philosophy is known as robust statistics (or, occasionally, as resistant statistics) and is expounded in a number of recent monographs,[7–10] some of which are forbidding even to professional statisticians.

Robust statistics have been used occasionally by chemists, especially in geochemistry.[11–15] These papers concentrate on establishing reference values, whereas robust methods can be as useful in assessing variability as for central tendency.

Philosophy of Robust Statistics

The normal distribution pervades statistical methodology, and its very name suggests widespread applicability. Yet careful studies show that real errors do not fit the normal distribution! Users of statistics point to a theoretical result, the central limit theorem, to justify the assumption, whereas theoreticians believe its applicability to have been proved empirically. The central limit theorem is, of course, a perfectly correct result about sums of many small independent errors having (approximately) a normal distribution. The problem is that outliers result from single large errors. It is also generally accepted that real error distributions have "heavier tails" than the normal distribution, *i.e.*, large deviations (in either direction) are more likely than under a normal distribution. One of the bases of robust statistics is to use procedures that work well for such distributions.

The second basis is to protect against gross errors. We observed from our example that recording 28.95 rather than 2.895 increased the sample mean considerably (to 4.28 from 3.19). Recording 289.5 and 2895 would give 15.1 and 123.7, respectively. Hence the effect of a missing decimal point is disastrous for the mean. On the other hand, the *median* is almost unchanged, from 3.24 with 2.895 to 3.38 with any value greater than 3.40. This property is shared by a *trimmed mean*. Suppose we discard the smallest r and the largest r observations out of the total, n, and then take the mean of the remainder. [This is called a $(100r/n)(\%)$ trimmed mean.] Discarding $r = 1$ and $r = 2$ gives estimates of the mean of 3.25 and 3.21, respectively, both being insensitive to the actual size of 28.95. Trimmed means obey both principles of robust statistics; they are insensitive to small numbers of gross errors, and they work well for heavy-tailed distributions close to the normal. The mean fails the first of these, the median the second.

The sample variance is even more sensitive to outliers than the sample mean, increasing from 0.46 (when 28.95 is replaced by 2.895) to 28.1. The inter-quartile range (IQR) is the difference between observations one quarter in from each end, the 6th and 19th in the present example, so IQR = 1.0. For a normal distribution the IQR would be expected to be about 1.35σ, which suggests that we determine σ^2 by $(\text{IQR}/1.35)^2 = 0.55$. Again this is insensitive to gross errors; it is in fact unchanged if 28.95 is replaced by 2.895 or 289.5.

The trimmed mean and the IQR were developed in the days when all data were analysed by hand. Computers have allowed more sophisticated methods to be used. Many such procedures exist, but we will only describe some of the simplest which are known to perform well. A whole book[16] has been devoted to comparisons of 68 procedures! The references disagree as to the best procedure, but all accept those described here as amongst the best.

Measuring "True Values"

Both the sample mean and median are estimates of the location of a distribution of results. This distribution can be considered as a "true value", μ, plus errors, and we want to find an estimate of μ. We assume that μ is the mean of the "reliable" results, but not of the whole error distribution, as in analytical chemistry the distribution of errors will almost always be asymmetric. Consider n data points $x_1, \ldots, x_n$. The sample mean minimises the sum of squares $SS = \Sigma(x_i - \mu)^2$ and this is the source of its sensitivity to gross errors as large errors inflate SS significantly. Suppose we minimise $SS = \Sigma\rho(x_i - \mu)$ where $\rho(\varepsilon)$ does not weight large errors, ε, as much as ε^2. A good choice is the function



Fig. 2. A robust loss function ρ. Note that $\rho(x) = x^2$ for $|x| \leqslant c$, but $\rho(x) < x^2$ for large $|x|$

$$\rho(\varepsilon) = \begin{cases} (\varepsilon/\sigma)^2, & |\varepsilon| \leqslant c\sigma \\ c(2|\varepsilon/\sigma| - c), & |\varepsilon| \geqslant c\sigma \end{cases}$$

illustrated in Fig. 2 where σ^2 is a robust variance and c is a constant in the range 1–2. This penalises errors larger than $c\sigma$ less severely than x^2. The corresponding location estimate, $\hat{\mu}$, is the mean of *pseudo-values* $\tilde{x}_i$

$$\tilde{x}_i = \begin{cases} x_i & \text{if } |x_i - \hat{\mu}| \leqslant c\sigma \\ \hat{\mu} - c\sigma & \text{if } x_i < \hat{\mu} - c\sigma \\ \hat{\mu} + c\sigma & \text{if } x_i > \hat{\mu} + c\sigma \end{cases} \quad \ldots \quad (2)$$

and also the weighted mean of x_i, with weights w_i

$$w_i = \begin{cases} 1 & \text{if } |x_i - \hat{\mu}| \leqslant c\sigma \\ c\sigma/|x_i - \hat{\mu}| & \text{if } |x_i - \hat{\mu}| > c\sigma \end{cases}$$

Hence extreme values can be thought of as being either brought in or downweighted. We can compute $\hat{\mu}$ from either of these properties. To start with take any estimate $\hat{\mu}^{(o)}$, say the mean or the median. At each stage compute $\hat{\mu}^{(j)}$ as the weighted mean with weights $\min(1, c\sigma/|x_i - \hat{\mu}^{(j-1)}|)$ or as the mean of the values $\tilde{x}_i$ [with $\hat{\mu} = \hat{\mu}^{(j-1)}$]. [The function $\min(x, y)$ denotes the smaller value of x and y.] The values $\hat{\mu}^{(j)}$ converge rapidly to $\hat{\mu}$.

The value of $c = 1.5$ has wide support. Suppose we knew σ to be 0.70, then starting from the mean

$$\hat{\mu}^{(j)} = 4.28, 3.56, 3.27, 3.22, 3.21, \ldots$$

and starting from the median

$$\hat{\mu}^{(j)} = 3.39, 3.24, 3.21, \ldots$$

It is usually unrealistic to assume that σ is known, although only a rough estimate is needed, which might be available from past trials. One rough estimate is based on the median absolute deviation (MAD); MAD = median $(|x_i - \text{median}|)$, $\hat{\sigma} = \text{MAD}/0.6745$, which is similar to the re-scaled IQR, and in our example gives $\hat{\sigma} = 0.53$ and $\hat{\mu} = 3.207$. (The scale factor 0.6745 is used to obtain the correct answer for normally distributed data.) More sophisticated estimates are considered later. However, this simple proposal is already very reliable. The corresponding estimator of location is sometimes known as A15.

Table 1. Values of the constants β and θ for a range of cut-off values c. Further values can be obtained from $\theta = P(|N| < c)$, $\beta = \theta + c^2(1 - \theta) - 2c\exp(-c^2/2)/\sqrt{2\pi}$ where N is a standard normal deviate

	c										
	1.0	1.1	1.2	1.3	1.4	1.5	1.6	1.7	1.8	1.9	2.0
β	0.516	0.578	0.635	0.688	0.736	0.778	0.816	0.849	0.877	0.900	0.921
θ	0.683	0.729	0.770	0.806	0.838	0.866	0.890	0.911	0.928	0.943	0.954

Measuring "Precision"

If it is supposed that we are in the somewhat more realistic position of knowing μ (say from a reference sample) and wish to estimate σ, then we could use the sample variance or the scaled IQR, or

$$\hat{\sigma}_m = \text{median}(|x_i - \mu|)/0.6745$$

A robust procedure is to solve

$$\Sigma \min(|x_i - \mu|/\sigma, c)^2 = n\beta \quad .. \quad .. \quad (3)$$

where again β is chosen to obtain the correct answer for normally distributed data. Some values of β are given in Table 1.

There are a number of ways to solve (3). One of the easiest is to compute a sequence of values $\hat{\sigma}^{(j)}$ with $\hat{\sigma}^{(o)} = \hat{\sigma}_m$ and

$$[\hat{\sigma}^{(j)}]^2 = \frac{1}{n\beta}\Sigma(\tilde{x}_i - \mu)^2$$

where

$$\tilde{x}_i = \begin{cases} x_i & \text{if } |x - \mu| < c\sigma^{(j-1)} \\ \mu - c\hat{\sigma}^{(j-1)} & \text{if } x_i < \mu - c\hat{\sigma}^{(j-1)} \\ \mu + c\hat{\sigma}^{(j-1)} & \text{if } x_i > \mu + \hat{c}\sigma^{(j-1)} \end{cases}$$

which converges rapidly to the solution of (3).

Suppose for dataset (1) that we knew that $\mu = 3.68$ (which is the consensus of a much larger set of measurements), then

$$\hat{\sigma}^i = 0.911, 0.927, 0.934, 0.938, 0.939, 0.940, 0.941, \ldots$$

so σ can be estimated by 0.941. Note that $\beta\hat{\sigma}^2$ is the variance of the pseudo-values $\tilde{x}_i$ (with divisor n as μ is known).

Unknown "true value"

The more sophisticated approach referred to in a previous section involves estimating σ alongside μ. Hence at each iteration we form pseudo-values $\tilde{x}_i$ and compute their mean, $\bar{x}$, and variance, s^2. Then $\hat{\mu}^{(i)} = \bar{x}$ and $\hat{\sigma}^{(i)} = \sqrt{(s^2/\beta)}$. This is repeated until the values stabilise, starting from (median, $\hat{\sigma}_m$). The present example gives

$\hat{\mu}$ =	3.385	3.255	3.213	3.206	3.205	3.205	...	3.205
$\hat{\sigma}$ =	0.526	0.595	0.639	0.657	0.666	0.671	...	0.674

The following shows the insensitivity to the outlier(s):

$x_{(24)}$	$\hat{\mu}$	$\hat{\sigma}$
28.95	3.205	0.674
2.895	3.146	0.613
289.5	3.205	0.674

and in fact $(\hat{\mu},\hat{\sigma})$ do not depend on any of the exact values greater than $\hat{\mu} + c\hat{\sigma} \approx 4.22$. Hence it is irrelevant whether the value 5.28 is considered an outlier or not; all that matters is that it exceeds 4.22. This combined estimator is known as H15 or "Huber proposal 2."

When n, the number of observations, is small, a small-sample correction should be made. The variance of $x_i - \hat{\mu}$ will be about $\sigma^2(n - 1)/n$ and so the cut-off point, c, should be reduced to $c\sqrt{(1 - 1/n)}$ in forming the pseudo-values. This will be important in Part 2.[17] (It reduces $\hat{\sigma}$ to 0.662 in our example.)

Discussion

We have observed that robust procedures can be constructed to estimate the true value and precision of a set of data by relatively simple iterative calculations. These are very tedious to do manually but easy to program. (The longest part of the program will be to find the starting values, see Appendix.) The robust estimates are completely insensitive to how outlying the extreme data values are and obtain most of their information from the values in the centre of the dataset.

The cut-off value c should in theory be chosen depending on how frequent outliers are thought to be, although it is safer to choose a smaller value of c if in doubt. About 1% of outliers suggest $c = 2.0$ and about 5% suggest $c = 1.4$. The value $c = 1.5$ is widely used. The actual estimates obtained are not very sensitive to c:

c =	1.0	1.5	2.0
$\hat{\mu}$ =	3.229	3.205	3.234
$\hat{\sigma}$ =	0.648	0.662	0.678

If we really wanted to look for outliers to check them against the original records, a useful rule would be to check x_i values outside $\hat{\mu} \pm 2\hat{\sigma}$. In dataset (1) this suggests that all values greater than 4.53 would be checked.

Use of robust estimates

Some considerable care is needed in interpreting $\hat{\mu}$ and $\hat{\sigma}$. They do *not* estimate the mean and standard deviation of the observations (note, *not* the population), and this is an asset rather than a liability. Rare but very large outliers will affect the theoretical mean μ considerably when, as in analytical chemistry, they will almost always occur in one direction. Instead we should regard μ as measuring the mean of the "reliable" observations, a consensus value which is the nearest we can get to a "true value". (This interpretation is only possible if the "reliable" observations form the majority. Examples do occur in which the outliers are the only valid observations, but no statistical procedure can redeem such a disastrous trial.)

In a similar manner $\hat{\sigma}$ measures the standard deviation of the "reliable" observations. If we take m replicates then the robust measure $\hat{\mu}$ obtained from these will have a variance of about $\hat{\sigma}^2/m$ for moderate m. [In fact $\hat{\sigma}^2/m \times \beta/\theta^2$ where $\theta = P(|x_i - \mu|/\sigma \leqslant c)$, where P is the probability, which can be estimated either from the normal distribution or by the proportion of the dataset with $\tilde{x}_i = x_i$. As Table 1 shows, the correction factor β/θ^2 is only just larger than one.] However the (population) variance of one observation will usually much exceed $\hat{\sigma}^2$, as outliers cannot be downweighted. Duplicates also do not help, as we always find that $\hat{\mu} = (x_1 + x_2)/2$. At least three replicates are needed to allow downweighting, and this may not be sufficient unless σ is known *a priori*. If we had recorded just the three observations 2.9, 3.1 and 28.95, then

Mean = 11.65	s = 14.98
Median = 3.1	$\hat{\sigma}_m$ = 0.297
A15 = 3.222	
H15 = 11.65	$\hat{\sigma}$ = 16.98

There are two plausible explanations for this triple of observations. One, favoured by (A15, $\hat{\sigma}_m$), is of two reliable

Program*

```
      program rob1
c
c       program from 'Robust Statistics - How Not to Reject Outliers'
c       Analyst
c       (C) B.D.Ripley
c
      parameter (NMAX = 100)
      real x(NMAX), ws(NMAX)
      real median, h15
      integer i, n
      real xmed, xm, xs
      character name*50
      print *, 'File name  '
      read '(a)', name
      open (1, file=name, status='old')
      print *, '# data points '
      read *, n
      read (1,*) (x(i), i=1,n)
      close (1)
      xmed = median(x, n, ws)
      print *, 'median ',xmed
      xm = a15(x, n, xs, ws)
      print *, 'a15, sigma ', xm, xs
      xm = h15(x, n, xs, ws)
      print *, 'h15, sigma ', xm, xs
      end

      real function median(x, n, ws)
      real x(n), ws(*), v
      integer i, j, h, n, n2
      do 10 i = 1, n
   10   ws(i) = x(i)
      h = 1
   20 h = 3*h+1
      if (h .le. n) goto 20
   30 h = h/3
      do 50 i = h+1, n
        v = ws(i)
        j = i
   40   if (ws(j-h) .le. v) goto 50
        ws(j) = ws(j-h)
        j = j-h
        if (j .gt. h) goto 40
   50   ws(j) = v
      if (h .gt. 1) goto 30
      n2 = n/2
      if (2*n2 .eq. n) then
        median = 0.5*(ws(n2)+ws(n2+1))
      else
        median = ws(n2+1)
      endif
      end

      real function smad(mu, x, n, ws)
      real median, mu, x(n), ws(*)
      real sm, sum
      integer i, n
      do 10 i = 1,n
   10   ws(i) = abs(x(i)-mu)
      sm = median(ws, n, ws)
      if (sm .le. 0.0) then
        sum = 0.0
        do 20 i = 1, n
   20     sum = sum + ws(i)
        sm = sum/n
      endif
      smad = sm/0.6745
      end

      real function a15(x, n, xs, ws)
      real median, x(n), ws(*)
      real c, xm, xm0, xs, xs0, xc, sum
      integer i, n
      data c/1.5/
      xm = median(x, n, ws)
      xs = smad(xm, x, n, ws)
      xc = c*xs
   10 xm0 = xm
      sum = 0.0
      do 20 i = 1, n
   20   sum = sum + min(xm0+xc, max(xm0-xc, x(i)))
      xm = sum/n
      if (abs(xm-xm0) .gt. (1.0e-4)*xs) go to 10
      a15 = xm
      end

      real function h15(x, n, xsc, ws)
      real median, x(n), ws(*), xsc
      real a, beta, c, c1, xm, xm0, xs, xs0, xc, sum, sum2
      integer i, n
      data c, beta /1.5, 0.778/
c     c1 = c   without small-sample correction
      c1 = c * sqrt(1.0-1.0/n)
      xm = median(x, n, ws)
      xs = smad(xm, x, n, ws)
   10 xm0 = xm
      xs0 = xs
      sum = 0.0
      sum2 = 0.0
      xc = c1*xs
      do 20 i = 1, n
        a = min(xm+xc, max(xm-xc, x(i)))
        sum = sum + a
   20   sum2 = sum2 + (a-xm)*(a-xm)
      xm = sum/n
      xs = sqrt(sum2/(beta*(n-1)))
      if ((abs(xm-xm0) .gt. 1.0e-4*xs0) .or.
     & abs(xs/xs0-1.0) .gt. 1.0e-4) go to 10
      h15 = xm
      xsc = xs
      end
```

* Copyright of Professor B. D. Ripley

observations plus one outlier. On the other hand, (H15, $\hat{\sigma}$) regards the sample as three variable observations. We know only from other data which explanation is correct.

No statistical method can make sense of disastrous trials. Abbey[14] quoted 31 determinations of the nickel content ($\mu g\ g^{-1}$) of Canadian syenite rock:

5.2 6.5 6.9 7 7 7 7.4 8 8 8 8 8.5 9 9 10
11 11 12 12 13.7 14 14 14 16 17 17 18 24 28 34 125

which gives

Mean = 16.01	s = 21.27
Median = 11.00	$\hat{\sigma}_m$ = 4.45
A15 = 11.55	
H15 = 11.70	$\hat{\sigma}$ = 5.19

suggesting a standard error of the robust mean of *ca.* 1.0 ($= \hat{\sigma}/\sqrt{m}$). Abbey quoted other robust estimators of $\hat{\mu}$, but all agreed to within the (considerable) uncertainty. However, this uncertainty is so large that very little has been learnt from 31 determinations.

In general, outliers are not a problem when data are looked at carefully. Increasingly data are not looked at at all. They are recorded in machine-readable form and summarised by computer programs. In such circumstances robust statistics are preferable to conventional ones, and a marked difference between them should give a warning that the data should be examined carefully.

Appendix

Computation

The exact form of the algorithms used to calculate robust estimates can be deduced from the FORTRAN 77 program shown. They cover the most general case of unknown μ and σ, but are easily modified to handle other instances.

Medians are found by sorting the data by the sort algorithm of Shell.[18] There are ways to find medians without sorting that will be faster for large values of n, but these are considerably more awkward to program correctly. One other difficulty is that $\hat{\sigma}_m$ could turn out to be zero, but only if half the data are equal to the median. In that instance we would report $\Sigma|x_i - \text{median}|/n$, which is zero only if *all* the data are equal to the median.

The programs use a common workspace (ws) which should be as large as the data array. One trap for the unwary: median and H15 are real qualities, despite the implicit rules of FORTRAN. To aid translation to other languages, all variables used are declared. NMAX can be set as required.

Professor Peter Rousseeuw's comments were most helpful in clarifying an earlier draft.

References

1. Dixon, W. J., *Ann. Math. Statist.*, 1950, **21**, 488.
2. Grubbs, F. E., *Technometrics*, 1969, **11**, 1.
3. Crow, E. L., Davis, F. A., and Maxfield, M. W., "Statistical Manual," Dover, New York, 1960.
4. "Precision of Test Methods," BS 1979, No. 5497, British Standards Institution, Milton Keynes.
5. Analytical Methods Committee, *Analyst*, 1989, **114**, 1489.
6. Barnett, V., and Lewis, J., "Outliers in Statistical Data," Second Edition, Wiley, Chichester, 1984.
7. Huber, P. J., "Robust Statistical Procedures," SIAM, Pennsylvania, 1976.
8. Huber, P. J., "Robust Statistics," Wiley, New York, 1981.
9. Hampel, F. R., Ronchetti, E. M., Rousseeuw, P. J., and Stahel, W. A., "Robust Statistics: The Approach Based on Influence Functions," Wiley, New York, 1986.
10. Rousseeuw, P. J., and Leroy, A. M., "Robust Regression and Outlier Detection," Wiley, New York, 1987.
11. Lister, B., *Geostand. Newsl.*, 1984, **8**, 171.
12. Lister, B., *Geostand. Newsl.*, 1987, **11**, 11.
13. Abbey, S., *Geostand. Newsl.*, 1986, **10**, 159.
14. Abbey, S., *Geostand. Newsl.*, 1988, **12**, 241.
15. Rock, N. M. S., *Math. Geol.*, 1988, **20**, 243.
16. Andrews, D. F., Bickel, P. J., Hampel, F. R., Huber, P. J., Rogers, W. H., and Tukey, J. W., "Robust Estimates of Location," Princeton University Press, 1972.
17. Analytical Methods Committee, *Analyst*, 1989, **114**, 1699.
18. Sedgewick, R., "Algorithms," Addison Wesley, Reading, MA, 1983.

Paper 9/02435B
Received June 9th, 1989